

## Microsoft Excel 2016

# Matrixfunktionen





## Inhaltsverzeichnis

Einleitung.....	2
Integrierte Matrixfunktionen.....	2
Verwendung einer Matrixfunktion .....	4
Matrixkonstante .....	5
Zellbereich einer Matrix markieren.....	6
Beispiele für Matrixfunktionen.....	7

## Abbildungsverzeichnis

Abb. 1: <i>Eine Matrixberechnung bzw. Matrixfunktion in der Bearbeitungsleiste.....</i>	4
Abb. 2: <i>Dialogfeld <b>Gehe zu</b> .....</i>	7
Abb. 3: <i>Dialogfeld <b>Inhalte auswählen</b>.....</i>	7
Abb. 4: <i>Beispiel einer „normalen Funktion“ als Matrixfunktion .....</i>	7
Abb. 5: <i>Beispiel für eine Matrixfunktion mit zweidimensionalen Matrizen .....</i>	8
Abb. 6: <i>Beispiel für eine Matrixkonstante .....</i>	8
Abb. 7: <i>Wiederholung von Beispiel 3 mit anderer Anordnung der Matrixkonstanten .....</i>	8
Abb. 8: <i>Beispiele für Matrixkonstanten.....</i>	9
Abb. 9: <i>Beispiel für eine Trendberechnung .....</i>	9
Abb. 10: <i>Das letzte Wort in einem Satz extrahieren .....</i>	10

## Einleitung

Eine Matrixformel kann mehrere Berechnungen durchführen und entweder ein einzelnes Ergebnis oder mehrere Ergebnisse liefern. Matrixformeln bearbeiten zwei oder mehr Wertsätze (Zellbereiche; keine einzelnen Tabellenzellen), die als Matrixargumente bezeichnet werden. Excel enthält spezielle Matrixfunktionen, aber es gibt auch Anwendungen, wo eine eigentlich „normale“ Formel als Matrixformel benutzt werden muss. Auch hierfür finden Sie in dem Skript ein Beispiel. Sie erstellen eine Matrixformel auf dieselbe Weise wie andere Formeln, aber mit dem Unterschied, dass Sie zur Bestätigung der Formeleingabe die Tastenkombination    verwenden. Der Inhalt dieses Skripts bezieht sich zwar in erster Linie auf Excel **2016** bzw. Excel für Office 365, gilt prinzipiell aber für alle Excel-Versionen (auch für Excel 2003 und älter).

## Integrierte Matrixfunktionen

Zunächst eine kleine Übersicht von Matrixfunktionen, die bereits bei Excel integriert sind (die Übersicht ist nicht unbedingt vollständig; die in der Farbe Grün angegebenen Argumente sind optional):

<b>Funktion</b>	<b>Beschreibung</b>
<b>BEREICH.VERSCHIEBEN</b>	Gibt einen Verweis auf einen Bereich zurück, der eine festgelegte Anzahl von Zeilen und Spalten von einer Tabellenzelle oder einem Zellbereich entfernt ist. Der zurückgegebene Verweis kann eine einzelne Tabellenzelle oder ein Zellbereich sein. Sie können die Anzahl der zurückzugebenden Zeilen und Spalten festlegen. <b>BEREICH.VERSCHIEBEN(Bezug;Zeilen;Spalten;Höhe;Breite)</b>
<b>BEREICHE</b>	Gibt die Anzahl der innerhalb eines Verweises aufgeführten Bereiche zurück. Ein Bereich (Teilbereich) kann sowohl aus mehreren zusammenhängenden Tabellenzellen (Zellbereich) als auch aus nur einer Tabellenzelle bestehen. <b>BEREICHE(Bezug)</b>
<b>HÄUFIGKEIT</b>	Berechnet, wie oft Werte innerhalb eines Wertebereichs auftreten, und gibt eine Häufigkeitsverteilung als einspaltige Matrix zurück. <b>HÄUFIGKEIT(Daten;Klassen)</b>
<b>INDEX</b>	Gibt den Wert eines Elements in einer Tabelle oder einer Matrix zurück. Das Element wird mit seiner Zeilen- und Spaltennummer angegeben. <b>INDEX(Matrix;Zeile;Spalte)</b> <b>INDEX(Bezug;Zeile;Spalte;Bereich)</b>
<b>KORREL</b>	Gibt den Korrelationskoeffizienten einer zweidimensionalen Zufallsgröße zurück, deren Werte in den Zellbereichen <b>Matrix1</b> und <b>Matrix2</b> stehen. Mit dem Korrelationskoeffizienten lässt sich feststellen, ob eine Beziehung zwischen zwei Eigenschaften vorliegt. <b>KORREL(Matrix1;Matrix2)</b>
<b>MDET</b>	Liefert die Determinante einer Matrix. <b>MDET(Matrix)</b>
<b>MINV</b>	Gibt die Inverse einer Matrix (die zu einer Matrix gehörende Kehrmatrix) zurück. <b>MINV(Matrix)</b>

<b>Funktion</b>	<b>Beschreibung</b>
<b>MMULT</b>	Gibt das Produkt zweier Matrizen zurück. Das Ergebnis ist eine Matrix, die dieselbe Anzahl an Zeilen wie <b>Matrix1</b> und dieselbe Anzahl an Spalten wie <b>Matrix2</b> aufweist. <b>MMULT(Matrix1;Matrix2)</b>
<b>MTRANS</b>	Mit dieser Funktion wird ein vertikaler Zellbereich als horizontaler Bereich zurückgegeben oder umgekehrt. Die Funktion muss als Matrixformel in einem Bereich eingegeben werden, der die gleiche Anzahl von Zeilen bzw. Spalten wie der Quellbereich hat. <b>MTRANS(Matrix)</b>
<b>PEARSON</b>	Gibt den pearsonschen Korrelationskoeffizienten r zurück. Dieser Koeffizient ist ein dimensionsloser Index mit dem Wertebereich zwischen -1,0 und 1,0 und ein Maß dafür, inwieweit zwischen zwei Datenmengen eine lineare Abhängigkeit besteht. <b>PEARSON(Matrix1;Matrix2)</b>
<b>RKP</b>	Liefert die Parameter eines exponentiellen Trends. In Regressionsanalysen berechnet diese Funktion eine Exponentialkurve, die möglichst gut an die von Ihnen bereitgestellten Daten angepasst ist, und liefert eine Matrix von Werten, die diese Kurve beschreibt. <b>RKP(Y_Werte;X_Werte;Konstante;Stats)</b>
<b>SCHÄTZER</b>	Gibt den Schätzwert für einen linearen Trend zurück. Der Vorhersagewert ist ein y-Wert für einen gegebenen x-Wert. <b>SCHÄTZER(x;Y_Werte;X_Werte)</b>
<b>SPALTE</b>	Mit dieser Funktion wird die Spaltennummer des jeweiligen Zellbezug zurückgegeben. <b>SPALTE(Bezug)</b>
<b>SPALTEN</b>	Gibt die Anzahl der Spalten eines Bezugs zurück. <b>SPALTEN(Matrix)</b>
<b>SUMMEX2MY2</b>	Summiert für sich entsprechende Werte zweier Matrizen die aus Quadratzahlen gebildeten Differenzen. <b>SUMMEX2MY2(Matrix_x;Matrix_y)</b>
<b>SUMMEX2PY2</b>	Summiert für sich entsprechende Werte zweier Matrizen die aus Quadratzahlen gebildeten Summen. <b>SUMMEX2PY2(Matrix_x;Matrix_y)</b>
<b>SUMMEXMY2</b>	Gibt die Summe der quadrierten Differenzen einander entsprechender Werte in zwei Matrix zurück. <b>SUMMEXMY2(Matrix_x;Matrix_y)</b>
<b>SVERWEIS</b>	Durchsucht die erste Spalte einer Matrix und durchläuft die Zeile nach rechts, um den Wert einer Tabellenzelle zurückzugeben. <b>SVERWEIS(Suchkriterium;Matrix;Spaltenindex;Bereich_Verweis)</b>
<b>TREND</b>	Gibt Werte zurück, die sich aus einem linearen Trend ergeben. <b>TREND(Y_Werte;X_Werte;Neue_X_Werte;Konstante)</b>
<b>VERGLEICH</b>	Gibt die relative Position eines Elements in einer Matrix zurück, die einem bestimmten Wert in einer angegebenen Reihenfolge entspricht. <b>VERGLEICH(Suchkriterium;Suchmatrix;Vergleichstyp)</b>

Funktion	Beschreibung
<b>VERWEIS</b>	Mit der Funktion wird ein Wert aus einem Bereich mit einer Zeile oder einer Spalte oder aus einer Matrix zurückgegeben. <b>VERWEIS(Suchkriterium;Suchvektor;Ergebnisvektor)</b> <b>VERWEIS(Suchkriterium;Matrix)</b>
<b>WVERWEIS</b>	Durchsucht die erste Zeile einer Matrix und durchläuft die Spalte nach unten, um den Wert einer Tabellenzelle zurückzugeben. <b>WVERWEIS(Suchkriterium;Matrix;Zeilenindex;Bereich_Verweis)</b>
<b>ZEILE</b>	Gibt die Zeilennummer eines Verweises zurück. <b>ZEILE(Bezug)</b>

Neben diesen Funktionen gibt es außerdem noch Befehle, die eine Matrixfunktion beinhalten. Dazu gehört beispielsweise die *Mehrfachoperation* (Register **Daten**, Gruppe **Datentools**, Symbol **Was-wäre-wenn-Analyse**, Befehl **Datentabelle**).



## Verwendung einer Matrixfunktion

Prinzipiell werden Matrixfunktionen genauso in Berechnungen verwendet wie „normale“ Funktionen. Trotzdem gibt es bei Matrixfunktionen auch Unterschiede. Hier drei, sehr bedeutsame Unterschiede:

1. Die meisten Matrixfunktionen liefern mehrere Ergebniswerte. Dazu müssen auch mehrere Tabellenzellen ausgewählt werden. Um wie viele Tabellenzellen es sich dabei genau handelt und ob dabei Tabellenzellen in einer Zeile oder Spalte (oder beides) markiert werden müssen, hängt von der jeweiligen Matrixformel und den Ausgangsdaten ab.
2. Wie fast alle Funktionen, benötigen auch Matrixfunktionen Angaben in Form von Funktionsargumenten. Bei diesen Funktionsargumenten handelt es sich in den meisten Fällen um Matrizen, was in Excel identisch ist mit Zellbereichen. Auch hier kommt es darauf an, ob die Daten in Zeilen- oder Spaltenform (oder beides) angeordnet sind.
3. Wird eine Matrixfunktion direkt in die Tabellenzelle(n) eingetragen, muss zur Eingabebestätigung die Tastenkombination  verwendet werden.

Wenn die Eingabe der Matrixfunktion mit der Tastenkombination  abgeschlossen worden ist, wird die gesamte Berechnung (inkl. dem Gleichheitszeichen) in geschweifte Klammern { und } eingeschlossen (sichtbar in der Bearbeitungsleiste; siehe Abbildung 1).



Die beiden geschweiften Klammern (hier rot eingefärbt) signalisieren, dass es sich um eine Matrixberechnung bzw. Matrixfunktion handelt.

Abb. 1: Eine Matrixberechnung bzw. Matrixfunktion in der Bearbeitungsleiste

**Anmerkung:** Wenn eine Matrixfunktion mehrere Ergebniswerte liefert, können Sie einen einzelnen Ergebniswert (oder einen Teil der Ergebniswerte) nicht überschreiben oder löschen. Dies geht nur wenn der Zellbereich mit allen Ergebniswerten markiert wird. Wenn Sie trotzdem versuchen, eine einzelne Ergebniszelle zu überschreiben oder zu löschen, können Sie den Versuch nur mit der Taste  abbrechen.

## Matrixkonstante

In einer gewöhnlichen Funktion können Sie entweder einen Bezug auf eine Tabellenzelle mit einem bestimmten Wert als Funktionsargument oder als konstanten Wert angeben. Entsprechend können Sie in einer Matrixformel, in der Sie einen Bezug auf einen Zellbereich verwenden, die Werte dieses Zellbereichs auch als konstante Matrix eingeben. Die Matrix der konstanten Werte wird als *Matrixkonstante* bezeichnet. Im Allgemeinen werden Sie Matrixkonstanten anstelle von Bezügen verwenden, wenn Sie vermeiden wollen, jede Konstante in eine eigene Tabellenzelle in Ihrer Tabelle einzugeben.

So geben Sie eine Matrixkonstante ein:

1. Geben Sie die Werte direkt in die Funktion ein, und schließen sie in geschweifte Klammern ein ({}).<sup>1</sup> Für die öffnende Klammer ( { ) verwenden Sie die Tastenkombination  und für die schließende Klammer ( } ) die Tastenkombination .
2. Trennen Sie Werte in verschiedenen **Spalten** durch **Punkte**.
3. Trennen Sie Werte in verschiedenen **Zeilen** durch **Semikola**.

Die nachfolgende Tabelle zeigt einige Beispiele von Matrixkonstanten. Dabei wird die Excel-Schreibweise die der mathematischen Schreibweise gegenübergestellt.

Beispiele:	Matrix-Schreibweise (Mathematik)	Excel-Schreibweise
	$\begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{pmatrix}$	{1.2.3;4.5.6}
	(10 20 30 40)	{10.20.30.40}
	$\begin{pmatrix} 32,6 \\ 54,9 \\ 12,7 \end{pmatrix}$	{32,6;54,9;12,7}
	$\begin{pmatrix} 1 & 2 & 3 \\ a & b & c \end{pmatrix}$	{1.2.3;"a"."b"."c"}
	$\begin{pmatrix} 56 & 34 \\ WAHR & FALSCH \end{pmatrix}$	{56.34;WAHR.FALSCH}

<sup>1</sup> Bitte verwechseln Sie diese geschweiften Klammern nicht mit denen, die sich am Anfang bzw. Ende der Matrixformeln befinden.

Matrixkonstanten können folgende Inhalte besitzen:


- Matrixkonstanten können Zahlen, Text, Wahrheitswerte oder Fehlerwerte enthalten.
- Zahlen in Matrixkonstanten können im Ganzzahlformat, Dezimalformat oder im wissenschaftlichen Zahlenformat (Exponentialschreibweise) angegeben werden.
- Text muss in doppelte Anführungszeichen (") eingeschlossen werden.
- In derselben Matrixkonstante können verschiedene Wertetypen gleichzeitig verwendet werden.
- Die Werte einer Matrixkonstante müssen Konstanten und dürfen keine Funktionen sein.
- Matrixkonstanten dürfen kein Dollarzeichen, Klammern oder Prozentzeichen enthalten.
- Matrixkonstanten dürfen keinen Zeilen oder Spalten ungleicher Länge enthalten.

Excel erlaubt es nicht, in einer Matrixfunktion Zellbezüge oder Namen genauso wie konstante Werte aufzulisten. Beispielsweise können Sie nicht mit **{A1.B1.C1}** die Matrix angeben, die die Werte aus den Tabellenzellen **A1**, **B1** und **C1** enthält. Stattdessen müssen Sie den Bezug auf den Zellbereich **A1:C1** verwenden oder die Werte dieser Tabellenzellen direkt als Konstanten in die Matrix eingeben, z.B. **{10.20.30}**.

## Zellbereich einer Matrix markieren

Wenn Sie in einer Arbeitsmappe viele Matrixberechnungen durchgeführt haben, ist es häufig schwierig nachzuvollziehen, in welchen Zellbereichen sich die Matrixberechnungen befinden. Sie können natürlich jede einzelne Tabellenzelle auswählen und in der Bearbeitungsleiste nachschauen, ob sich in der Tabellenzelle eine Matrixformel befindet oder nicht (erkennbar an den geschweiften Klammern vor und hinter der Formel; siehe Abbildung 1, Seite 4). Allerdings wissen Sie dann immer noch nicht, wie groß der zugehörige Zellbereich ist, in dem sich die Matrixformel befindet. Excel stellt aber zwei Möglichkeiten bereit, mit denen Sie den kompletten Zellbereich einer Matrixformel markieren lassen können. Voraussetzung ist allerdings, dass mindestens eine Tabellenzelle der Matrixformel ausgewählt ist:

- Wählen Sie im Register **Start** in der Gruppe **Bearbeiten** das Symbol **Suchen und Auswählen** und den Befehl **Gehe zu** (alternativ benutzen Sie die Funktionstaste ). Im Dialogfeld **Gehe zu** (siehe Abbildung 2, Seite 7) klicken Sie auf die Schaltfläche . Im Dialogfeld **Inhalte auswählen** wählen Sie die Option **Aktuelles Array** (siehe Abbildung 3, Seite 7) und bestätigen das Dialogfeld.
- Benutzen Sie die Tastenkombination   .






Abb. 2: Dialogfeld **Gehe zu**




Abb. 3: Dialogfeld **Inhalte auswählen**

## Beispiele für Matrixfunktionen

Im ersten Beispiel<sup>2</sup> wird eine „normale“ Funktion als Matrixfunktion eingesetzt. Es handelt sich in diesem Beispiel um die Funktion **SIN**. Dabei soll der Sinus für die X-Werte von **-2,0** bis **+2,0** (in Schritten von **0,5**) ermittelt werden (siehe Abbildung 4).

	A	B	C	D	E	F	G	H	I	J
1	x	fx=sin(x)	fx=sin(x)							
2	-2,0	-0,9093	-0,9093		In die Tabellenzelle <b>B2</b> wird die Formel <b>=SIN(A2)</b> eingetragen und ganz normal mit der Eingabetaste (<Return>) bestätigt. Danach Tabellenzelle <b>B2</b> auswählen und die Ergebnisse für die Tabellenzellen <b>B3:B10</b> mit dem automatischen Ausfüllen bis zur Tabellenzelle <b>B10</b> kopieren.					
3	-1,5	-0,9975	-0,9975							
4	-1,0	-0,8415	-0,8415							
5	-0,5	-0,4794	-0,4794							
6	0,0	0,0000	0,0000							
7	0,5	0,4794	0,4794		Zunächst wird der Zellbereich <b>C2:C10</b> markiert und anschließend die Formel <b>=SIN(A2:A10)</b> eingetragen. Für die Berechnung der Ergebnisse muss nun noch die Tastenkombination <Strg>+<Umschalten>+<Eingabetaste> gedrückt werden.					
8	1,0	0,8415	0,8415							
9	1,5	0,9975	0,9975							
10	2,0	0,9093	0,9093							

Abb. 4: Beispiel einer „normalen Funktion“ als Matrixfunktion

Das zweite Beispiel zeigt eine Matrixfunktion, die eine zweidimensionale Matrix als Ausgangsmatrix benötigt und die eine zweidimensionale Matrix als Ergebnis liefert. Für dieses Beispiel wurde die Funktion **MINV** (invertierte Matrix) genommen (siehe Abbildung 5, Seite 8).

<sup>2</sup> Die Angabe <Strg>+<Umschalten>+<Eingabetaste> in den einzelnen Beispielen entspricht der Tastenkombination



	A	B	C	D	E	F	G	H	I
1		34	65	23			-0,041	0,163	0,020
2	Matrix A =	5	8	3		MINV(A) =	0,224	-1,898	0,024
3		77	66	22			-0,531	5,122	-0,098
4									
5							Zunächst wird der Zellbereich G1:I3 markiert, dann die Formel =MINV(B1:D3) eingetragen und zum Abschluß mit der Tastenkombination <Strg>+<Umschalten>+<Eingabetaste> bestätigt.		
6									
7									
8									

Abb. 5: Beispiel für eine Matrixfunktion mit zweidimensionalen Matrizen

Im dritten Beispiel werden die Elemente zweier Matrizen multipliziert (Funktion **MMULT**). Dabei wird die eine Matrix in einem Zellbereich angegeben, während die zweite Matrix in Form einer Matrixkonstante in die Formel eingetragen wird (siehe Abbildung 6).

	A	B	C	D	E	F	G	H	I	J	K				
1		1	10	100			963								
2	Matrix A =	5	50	500			4815								
3		9	90	900			8667								
4															
5		$A * \begin{pmatrix} 3 \\ 6 \\ 9 \end{pmatrix} = \begin{pmatrix} 1 & 10 & 100 \\ 5 & 50 & 500 \\ 9 & 90 & 900 \end{pmatrix} * \begin{pmatrix} 3 \\ 6 \\ 9 \end{pmatrix}$					Zunächst wird der Zellbereich G1:G3 markiert, dann die Formel =MMULT(B1:D3;{3;6;9}) eingetragen und mit der Tastenkombination <Strg>+<Umschalten>+<Eingabetaste> abgeschlossen.								
6															
7															
8															
9	Im grünen Bereich zunächst die mathematische Formel für die														
10	Matrixmultiplikation. Die Matrix A ist vorgegeben, während die														
11	zweite Matrix als Matrixkonstante in die Formel eingegeben														
12	wird.														

Abb. 6: Beispiel für eine Matrixkonstante

Im vierten Beispiel wird noch mal Beispiel 3 wiederholt. Diesmal allerdings wird bei der Matrixkonstanten nicht das Semikolon als Trennzeichen zwischen den Werten **3**, **6** und **9** genommen, sondern der Punkt (siehe Abbildung 7).

	A	B	C	D	E	F	G	H	I	J					
1		1	10	100			#WERT!								
2	Matrix A =	5	50	500			#WERT!								
3		9	90	900			#WERT!								
4															
5		$A * \begin{pmatrix} 3 \\ 6 \\ 9 \end{pmatrix} = \begin{pmatrix} 1 & 10 & 100 \\ 5 & 50 & 500 \\ 9 & 90 & 900 \end{pmatrix} * \begin{pmatrix} 3 \\ 6 \\ 9 \end{pmatrix}$					Zunächst wird der Zellbereich G1:G3 markiert, dann die Formel =MMULT(B1:D3;{3.6.9}) eingetragen und mit der Tastenkombination <Strg>+<Umschalten>+<Eingabetaste> bestätigt.								
6															
7															
8		Das Ergebnis ist allerdings diesmal ein Fehlerwert, denn die Matrixkonstante wurde als 1x3-Matrix (Matrixelemente werden jeweils durch einen Punkt getrennt) und nicht als 3x1-Matrix (Matrixelemente werden jeweils durch ein Semikolon getrennt) angegeben.													
9		Dieses Beispiel zeigt wie wichtig es ist, den richtigen Zellbereich für die Matrixberechnung anzugeben.													
10															
11															
12															
13															

Abb. 7: Wiederholung von Beispiel 3 mit anderer Anordnung der Matrixkonstanten

Im fünften Beispiel wird nur eine Matrixkonstante verwendet. Als Beispiel wird eine transponierte Matrix mit der Funktion **MTRANS** erzeugt (siehe Abbildung 8, Seite 9).

	A	B	C	D	E	F	G	H	I	J
1	1	4	7		Zellbereich <b>A1:C3</b> markieren, dann Formel <b>=MTRANS({1.2.3;4.5.6;7.8.9})</b> eingeben und mit der Tastenkombination <b>&lt;Strg&gt;+&lt;Umschalten&gt;+ &lt;Eingabetaste&gt;</b> bestätigen.					
2	2	5	8							
3	3	6	9							
4										
5	1	4			Zellbereich <b>A5:B7</b> markieren, dann Formel <b>=MTRANS({1.2.3;4.5.6})</b> eingeben und mit der Tastenkombination <b>&lt;Strg&gt;+&lt;Umschalten&gt;+&lt;Eingabetaste&gt;</b> bestätigen.					
6	2	5								
7	3	6								
8										
9	1	3			Zellbereich <b>A9:B11</b> markieren, dann Formel <b>=MTRANS({1.2;3.4;5.6})</b> eingeben und mit der Tastenkombination <b>&lt;Strg&gt;+&lt;Umschalten&gt;+&lt;Eingabetaste&gt;</b> bestätigen.					
10	2	4								
11	#NV	#NV								
12					Bei diesem Beispiel entsteht in den Tabellenzellen <b>A11</b> und <b>B11</b> jeweils ein Fehlerwert. Für diese Formel hätte eigentlich der Zellbereich <b>A9:C10</b> markiert werden müssen.					
13										
14										

Abb. 8: Beispiele für Matrixkonstanten

Im sechsten Beispiel wird die Matrixfunktion **TREND** gezeigt, die eher einen Bezug zur Praxis hat als die Funktionen **MINV**, **MMULT** oder **MTRANS** aus der linearen Algebra. Ausgehend von einer Grundmenge an existierenden Werten berechnet die Funktion **TREND** eine Reihe von Werten, die, wenn Sie in einem XY-Diagramm als Linie dargestellt werden, eine steigende oder sinkende Gerade ergibt.

	A	B	C	D	E	F	G
1		Jan	Feb	Mrz	Apr	Mai	Jun
2	Verkaufszahlen	10.765	11.263	11.039	11.471	11.832	11.621
3	Trend	10.873	11.057	11.240	11.424	11.607	11.790
4							
5	Zellbereich <b>B3:G3</b> markieren, dann die Formel <b>=TREND(B2:G2)</b> eingeben und mit der Tastenkombination <b>&lt;Strg&gt;+&lt;Umschalten&gt;+&lt;Eingabetaste&gt;</b> bestätigen.						
6							
7	Die Ergebnisse wurden mit dem Zahlenformat <b>##0</b> (ohne Dezimalstellen) formatiert.						

Abb. 9: Beispiel für eine Trendberechnung

Bei dem siebten Beispiel handelt es sich auf den ersten Blick gar nicht um eine Matrixformel, allerdings muss sie als Matrixformel bestätigt werden, damit das gewünschte Ergebnis herauskommt. Dabei geht es in der Formel darum, aus einem längeren Satz, der aus mehreren Wörtern besteht, das letzte Wort zu extrahieren. Als Beispiel steht der Satz in der Tabellenzelle **A1**<sup>3</sup> und die Formel in der Tabellenzelle **B1**. Die Formel lautet:

**=TEIL(A1;MAX((TEIL(A1;ZEILE(INDIREKT("1:"&LÄNGE(A1)))));1)=" ")\*ZEILE(INDIREKT("1:"&LÄNGE(A1)))+1;LÄNGE(A1))**

Die Formel sieht recht kompliziert aus und sie ist es auch. An dieser Stelle wird auf eine genaue Beschreibung der einzelnen Schritte der Formel verzichtet. Vertrauen Sie einfach darauf, dass die Formel genau das tut, was sie soll. Nach der Formeleingabe müssen Sie die Eingabe mit der Tastenkombination abschließen, auch wenn es sich gar nicht um Matrixfunktionen handelt und die Formel auch nur in eine einzige Tabellenzelle eingetragen worden ist. Bestätigen Sie die Formeleingabe nur mit der Eingabetaste, erhalten Sie ein falsches Ergebnis. Abbildung 10, Seite 10, zeigt nochmal die Formel in Aktion.

<sup>3</sup> Der Satz dürfte sicherlich breiter sein als die Spalte, in der er steht. Aber es ist ja kein großes Problem, die Spaltenbreite an die Satzlänge anzupassen.

	A	B	C	D	E	F
1	Das ist ein langer Satz.	Satz.				
2	Das ist ein ganz langer Satz ohne Punkt und Komma	Komma				
3						
4	Geben Sie in die Tabellenzelle B1 die Formel					
5						
6	<code>=TEIL(A1;MAX((TEIL(A1;ZEILE(INDIREKT("1:"&amp;LÄNGE(A1)));1)=" ")*ZEILE(INDIREKT("1:"&amp;LÄNGE(A1))))+1;LÄNGE(A1))</code>					
7						
8	ein und bestätigen die Eingabe mit <Strg>+<Umschalten>+<Eingabetaste>.					

Abb. 10: Das letzte Wort in einem Satz extrahieren

**Anmerkung:** Sie werden sich vielleicht fragen, muss das so kompliziert sein, oder geht das nicht auch mit einer einfacheren Formel? Die Antwort ist: ja, es gibt eine einfachere Formel, die auch nicht als Matrixformel eingegeben werden muss. Die gezeigte Formel soll ja auch nur als Beispiel dafür dienen, dass es Situationen in Excel gibt, wo Sie zur Lösung eines Problems eine Matrixformel benötigen, auch wenn die Formel keine Matrixfunktionen enthält und sie auch nur einen Ergebniswert liefert. Übrigens: die einfachere Formel lautet (auch wenn sie immer noch recht lang ist und kompliziert aussieht):

```
=RECHTS(A1;LÄNGE(A1)-FINDEN("#";WECHSELN(A1;" ";"#";LÄNGE(A1)-LÄNGE(WECHSELN(A1;" "))))
```

Wie Sie am ersten Beispiel in Abbildung 10 sehen können, liefert die Formel (es spielt keine Rolle, ob es sich hierbei um die Matrixformel oder die in dieser Anmerkung gezeigten kürzeren Formel handelt) beim Ergebnis auch noch den Punkt hinter dem Wort. Die Formeln sind also nicht ganz perfekt.